# IVRG, POLYNOMIALS AND SYMMETRY
## $g = 5,\ G = \langle 192, 181\rangle$

TYLER JOHNSON

The group $G = \langle 192, 181\rangle$ is the automorphism group of a genus 5 curve [2]. We use `DecomposeGAction`, in conjunction with the Chevalley-Weil and Eichler Trace formulas, to find equations for a curve with this automorphism group.

First, we find matrix generators for the action of $\mathrm{Aut}(C)$ on the vector space $H^0(C, K)$. These are given in [1], Prop. 3.6, p. 92.

Let $z = e^{2\pi i/8}$. Then the generators are

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & i & 0 & 0 & 0 \\ 0 & 0 & i & 0 & 0 \\ 0 & 0 & 0 & -i & 0 \\ 0 & 0 & 0 & 0 & -i \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} -1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & -1 \end{pmatrix},$$

$$\begin{pmatrix} -i & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & i & 0 \\ 0 & 0 & i & 0 & 0 \\ 0 & i & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 & 1 \\ 0 & z^5/\sqrt{2} & 0 & z^5/\sqrt{2} & 0 \\ 1 & 0 & 0 & 0 & 0 \\ 0 & z^7/\sqrt{2} & 0 & z^3/\sqrt{2} & 0 \\ 0 & 0 & 1 & 0 & 0 \end{pmatrix}, \begin{pmatrix} -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & z & 0 \\ 0 & 0 & 0 & 0 & -1 \\ 0 & z^7 & 0 & 0 & 0 \\ 0 & 0 & -1 & 0 & 0 \end{pmatrix}$$

which we call $A$, $B$, $C$, $D$, $E$, and $F$ in the `Magma` session below.

```
> K<z>:=CyclotomicField(8);
> i:=z^2;
> sqrt2:=z+z^7;
> sqrt2^2;
2
> GL5K:=GeneralLinearGroup(5,K);
> A:=elt<GL5K | 1,0,0,0,0, 0,i,0,0,0, 0,0,i,0,0, 0,0,0,-i,0, 0,0,0,0,-i>;
> B:=elt<GL5K | 1,0,0,0,0, 0,1,0,0,0, 0,0,-1,0,0, 0,0,0,1,0, 0,0,0,0,-1>;
> C:=elt<GL5K | -1,0,0,0,0, 0,1,0,0,0, 0,0,1,0,0, 0,0,0,1,0,
0,0,0,0,-1>;
> D:=elt<GL5K | -i,0,0,0,0, 0,0,0,i,0, 0,0,i,0,0, 0,i,0,0,0,
0,0,0,0,1>;
> E:=elt<GL5K | 0,0,0,0,1, 0,z^5/sqrt2,0,z^5/sqrt2,0, 1,0,0,0,0,
0,z^7/sqrt2,0,z^3/sqrt2,0, 0,0,1,0,0>;
> F:=elt<GL5K | -1,0,0,0,0, 0,0,0,z,0, 0,0,0,0,-1, 0,z^7,0,0,0, 0,0,-1,0,0>;
> G:=sub<GL5K | A,B,C,D,E,F>;
> IdentifyGroup(G);
<192,181>
```

$E$ requires the use of square root of two; in $\mathbb{Q}[\zeta_8]$, the square root of two can be written as $\zeta_8 + \zeta_8^7$.

```
> load "DGAv3.txt";
Loading "DecomposeGAction.txt"
```

*Date*: November 15, 2010.

```
> S<a,b,c,d,e>:=PolynomialRing(K,5);
> DecomposeGAction(G,S,2);
[
    rec<recformat<CharacterRow, Dimension, Multiplicity, Elements> |
        CharacterRow := 6,
        Dimension := 6,
        Multiplicity := 2,
        Elements := [
            a^2,
            b^2,
            b*d,
            c^2,
            d^2,
            e^2
        ]>,
    rec<recformat<CharacterRow, Dimension, Multiplicity, Elements> |
        CharacterRow := 11,
        Dimension := 3,
        Multiplicity := 1,
        Elements := [
            a*c,
            a*e,
            c*e
        ]>,
    rec<recformat<CharacterRow, Dimension, Multiplicity, Elements> |
        CharacterRow := 15,
        Dimension := 6,
        Multiplicity := 1,
        Elements := [
            a*b,
            a*d,
            b*c,
            b*e,
            c*d,
            d*e
        ]>
]
```

It is not clear from `DecomposeGAction` where our polynomials lie, so we turn to our `Magma` implementations of the Chevalley-Weil and Eichler Trace formulas (whose commands are `CW` and `Eichler`, respectively).

First, `Eichler` finds a set of surface kernel generators.

```
> load "eichlerv3.txt";
Loading "eichlerv3.txt"
> load "CWv2.txt";
Loading "CWv2.txt"
> SKG:=AllSurfaceKernelGenerators(G,[2,3,8]);
> #SKG;
384
> chi:=Character(GModule(G));
```

```
> L:=[ chi eq Eichler(G,5,SKG[i]) : i in [1..10]];
> L;
[ false, true, false, false, false, true, true, false, false, false ]
```

We have `Eichler` test to the first ten of 384 sets of surface kernel generators to see if any are compatible with our set of matrix generators for $G$; the second turns out to be.

Now, we can work with `CW`.

```
> M:=SKG[2];
> T:=CharacterTable(G);
> CCL:=Classes(G);
> CW(G,0,T,CCL,M,2,S);
S_m=
[
    0,
    0,
    0,
    0,
    0,
    2,
    0,
    0,
    0,
    0,
    1,
    0,
    0,
    1,
    0
]


H^0(C,mK)=
[ 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 0, 0, 1, 0 ]


I_m=
[
    0,
    0,
    0,
    0,
    0,
    1,
    0,
    0,
    0,
    0,
    0,
    0,
    0,
```

```
    0,
    0
]
```

The values of $I_m$ tell us in which character row of $G$ we should be looking for our equations; it turns out to be the sixth character row. Referring back to the output of `DecomposeGAction`, we see our equations must lie in $\mathrm{Span}\{a^2, b^2, bd, c^2, d^2, e^2\}$.

In order to figure out what type of equations we are looking for, we must know whether or not $G$ is hyperelliptic; we can find this with a command from `Eichler`, `IsHyperelliptic`, using the set of surface kernel generators we selected earlier, $M$.

```
> IsHyperelliptic(G,5,M);
false
```

Let us assume $G$ is not trigonal; we will attempt to verify this by finding equations consistent with a general group. So, by this assumption, we are looking for a 3-dimensional subspace of $\mathrm{Span}\{a^2, b^2, bd, c^2, d^2, e^2\}$.

Peering a little further into `DecomposeGAction`, we learn more about the this three dimensional subspace.

```
> S2,i2,B2:=GModule(G,S,2);
> V6:=sub<S2 | i2(a^2),i2(b^2),i2(b*d),i2(c^2),i2(d^2),i2(e^2)>;
> E:=EndomorphismRing(V6);
> Image(E.1);
Vector space of degree 6, dimension 3 over K
Echelonized basis:
(1 0 0 0 0 0)
(0 0 0 1 0 0)
(0 0 0 0 0 1)
> Image(E.2);
Vector space of degree 6, dimension 3 over K
Echelonized basis:
(0 1 0 0 0 0)
(0 0 1 0 0 0)
(0 0 0 0 1 0)
> Image(E.3);
Vector space of degree 6, dimension 3 over K
Echelonized basis:
(0 1 0 0 0 0)
(0 0 1 0 0 0)
(0 0 0 0 1 0)
> Image(E.4);
Vector space of degree 6, dimension 3 over K
Echelonized basis:
(1 0 0 0 0 0)
(0 0 0 1 0 0)
(0 0 0 0 0 1)
```

The bases of the images of the endomorphism ring tell us we need something more specific than a $G$-invariant, 3-dimensional subspace of $\mathrm{Span}\{a^2, b^2, bd, c^2, d^2, e^2\}$; we actually need a subspace of $\mathrm{Span}\{a^2, c^2, e^2\} + \mathrm{Span}\{b^2, bd, d^2\}$.

To show a putative subspace is $G$-invariant, is sufficient to show that it is invariant under our set of surface kernel generators, $M$, the elements of which are

These matrices have some incorrect signs

$$\begin{pmatrix} 0 & 0 & -1 & 0 & 0 \\ 0 & -1/\sqrt{2} & 0 & -i/\sqrt{2} & 0 \\ -1 & 0 & 0 & 0 & 0 \\ 0 & i/\sqrt{2} & 0 & 1/\sqrt{2} & 0 \\ 0 & 0 & 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & i & 0 & 0 \\ 0 & z^3/\sqrt{2} & 0 & z^5/\sqrt{2} & 0 \\ 0 & 0 & 0 & 0 & -1 \\ 0 & z^7/\sqrt{2} & 0 & z^5/\sqrt{2} & 0 \\ i & 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 0 & i \\ 0 & i/\sqrt{2} & 0 & i/\sqrt{2} & 0 \\ 0 & 0 & i & 0 & 0 \\ 0 & -i/\sqrt{2} & 0 & i/\sqrt{2} & 0 \\ 1 & 0 & 0 & 0 & 0 \end{pmatrix}$$

(referred to as $M[1]$, $M[2]$, and $M[3]$). However, since $M$ is a set of surface kernel generators, we have $M[1] * M[2] * M[3] = Id$; therefore $M[3] = (M[1] * M[2])^{-1}$, so invariance under $M[3]$ follows directly from invariance under $M[1]$ and $M[2]$.

So, beginning with Span$\{a^2, c^2, e^2\}$, under $M[1]$, $a^2 \mapsto c^2$, $c^2 \mapsto a^2$, $e^2 \mapsto e^2$, and under $M[2]$, $a^2 \mapsto -c^2$, $c^2 \mapsto e^2$, and $e^2 \mapsto -a^2$. We need to pair these with three elements of Span$\{b^2, bd, d^2\}$ that have the same action under $M[1]$ and $M[2]$, i.e. $\alpha$, $\beta$, and $\gamma \in$ Span$\{b^2, bd, d^2\}$ such that, under $M[1]$, $\alpha \mapsto \beta$, $\beta \mapsto \alpha$, and $\gamma \mapsto \gamma$, and likewise for $M[2]$; such $\alpha$, $\beta$, and $\gamma$ constitute a basis for Span$\{b^2, bd, d^2\}$, and ensure invariance of our polynomial.

$\alpha = -2ibd$, $\beta = b^2 + d^2$, and $\gamma = ib^2 - id^2$ are some such elements of Span$\{b^2, bd, d^2\}$, so our the equations for $G$ are $a^2 - 2ibd$, $c^2 + b^2 + d^2$, and $e^2 + ib^2 - id^2$, if these equations are nonsingular, which we can verify with `Magma`.

```
> P4<a,b,c,d,e>:=ProjectiveSpace(K,4);
> X:=Scheme(P4,[a^2-2*i*b*d, c^2+b^2+d^2, e^2+i*b^2-i*d^2]);
> IsNonsingular(X);
true
```

## References


[1] Akikazu Kuribayashi and Hideyuki Kimura, *Automorphism groups of compact Riemann surfaces of genus five*, J. Algebra **134** (1990), no. 1, 80–103, DOI 10.1016/0021-8693(90)90212-7. MR1068416 (91j:30033) ←1

[2] K. Magaard, T. Shaska, S. Shpectorov, and H. Völklein, *The locus of curves with prescribed automorphism group*, Sūrikaisekikenkyūsho Kōkyūroku **1267** (2002), 112–141, available at arXiv:math.AG/0205314. Communications in arithmetic fundamental groups (Kyoto, 1999/2001). MR1954371 ←1


## Software Packages Referenced


[3] School of Mathematics and Statistics Computational Algebra Research Group University of Sydney, MAGMA *computational algebra system* (2008), available at http://magma.maths.usyd.edu.au/magma/. Version 2.15-1. ←